



プログラミング演習 (9)

多重配列

中村, 小松, 渡邊
小林, 鹿喰, 崎峨山




- Processing で多重配列に挑戦！
 - 2次元のマス目に配置されたオブジェクトをどう扱っていくか？
- 課題：
 - オセロゲームを作ってみる
 - ライツアウトを作ってみよう

2次元配列の定義



int[][] square = new int [10][5];


- 整数型で要素数が 10×5 個の square という配列を作成
- square の中に小箱が 10×5 できているイメージ



Tic Tac Toe





- Tic Tac Toe(マルバツゲーム)を作つてみよう！
 - ○と×の変わりに、白丸と黒丸で実現しよう！
 - クリックする度に【何も置かれていない→白が置かれている
→黒が置かれている】と切り替える
 - どうやってマス目に何が置かれているかを管理する？
 - square という配列を用意し
 - 何も置かれていないか
 - 黒が置かれているか
 - 白が置かれているか
- を管理する



どう考えるか？



- まずはマス目を描画してみよう
 - line を4本引くことで描画することが可能
- 次に、1つずつのマス目を配列で定義しよう
 - square[3][3] という配列で管理する！
 - 最初に配列を0で初期化し、配列の値が0なら空白、1なら白色の丸、2なら黒色の丸を描画する。



まず初期化！



- 2次元配列の定義

変数の型 [][] 変数名 = new 変数の型 [要素数][要素数];

```
int [][] square = new int [3][3];
```

while

```
void setup(){  
    size( 300, 300 );  
    int x=0;  
    while( x<3 ){  
        int y=0;  
        while( y<3 ){  
            square[x][y] = 0;  
            y++;  
        }  
        x++;  
    }  
}
```

3x3の2次元配列の定義

for

```
int [][] square = new int [3][3];
```

```
void setup(){  
    size( 300, 300 );  
    for( int x=0; x<3; x++ ){  
        for( int y=0; y<3; y++ ){  
            square[x][y] = 0;  
        }  
    }  
}
```

多重ループで0に初期化

描画する！



- square[x][y] の値に応じて描画
 - 多重繰り返しで描画する(線の描画は省略)

```
void draw(){  
    background( 255 );  
    // ここで線を描画しておく  
    int x=0;  
    while( x<3 ){  
        int y=0;  
        while( y<3 ){  
            if( square[x][y] == 0 ){  
                // 何も描画しない  
            } else if( square[x][y] == 1 ){  
                // 白丸を描画する  
            } else if( square[x][y] == 2 ){  
                // 黒丸を描画する  
            }  
            y++;  
        }  
        x++;  
    }  
}
```

while

```
void draw(){  
    background( 255 );  
    // ここで線を描画しておく  
    for( int x=0; x<3; x++ ){  
        for( int y=0; y<3; y++ ){  
            if( square[x][y] == 0 ){  
                // 何も描画しない  
            } else if( square[x][y] == 1 ){  
                // 白丸を描画する  
            } else if( square[x][y] == 2 ){  
                // 黒丸を描画する  
            }  
        }  
    }  
}
```


for

マウスクリックでの変化

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- マウスクリックされた時に、そのクリックされた座標に応じて、どの配列の値を変更するかを考える！
 - 1マスが100ピクセルなので、`mouseX / 100`や、`mouseY / 100`の値(値を切り捨てたもの)が配列の添え字([]の中で指定する値)となる




```
void mousePressed(){  
    // どこのマス目をクリックしたか  
    // マス目の大きさが100なので...  
    int tx = mouseX / 100;  
    int ty = mouseY / 100;  
    square[tx][ty]++;  
    if( square[tx][ty] > 2 ){  
        square[tx][ty] = 0;  
    }  
}
```

Tic Tac Toe 改

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- Tic Tac Toe(マルバツゲーム)を作つてみよう！
 - ○と×の変わりに、白丸と黒丸で実現しよう！
 - クリックするだけで白丸の後は黒丸、黒丸の後は白丸を自動で続けるようにせよ
 - 考え方
 - 初期化とかは問題ない
 - ターンという考え方を導入し、ターンに応じて置くコマを切り替える
 - Tic Tac Toe のmousePressedだけを変更！



Tic Tac Toe 改



- turn が0ならsquare[tx][ty]には1を, turnが1ならsquare[tx][ty]には2をセットする！
- クリックの度にturnの値を変化させる！

```
int [][] square = new int [3][3];
int turn = 0;
void setup(){
    size( 300, 300 );
    int x=0;
    while( x<3 ){
        int y=0;
        while( y<3 ){
            square[x][y] = 0;
            y++;
        }
        x++;
    }
}
```

while

```
void mousePressed(){
    int tx = mouseX / 100;
    int ty = mouseY / 100;
    if( turn == 0 ){
        square[tx][ty] = 1;
    } else {
        square[tx][ty] = 2;
    }
    turn++;
    if( turn > 1 ){
        turn = 0;
    }
}
```

for

もっと簡単に
できないか？

Tic Tac Toe 改



- turn を1と2が変化する変数にする
 - turnが1ならsquare[tx][ty]には1を, turnが2ならsquare[tx][ty]には2をセットする！つまり turnを代入するだけで良い！

```
int [][] square = new int [3][3];
int turn = 1;
void setup(){
    size( 300, 300 );
    int x=0;
    while( x<3 ){
        int y=0;
        while( y<3 ){
            square[x][y] = 0;
            y++;
        }
        x++;
    }
}
```

while

```
void mousePressed(){
    int tx = mouseX / 100;
    int ty = mouseY / 100;
    square[tx][ty] = turn;

    turn++;
    if( turn > 2 ){
        turn = 1;
    }
}
```


for

Tic Tac Toe 改二

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- Tic Tac Toe(マルバツゲーム)を作つてみよう！
 - ○と×の変わりに、白丸と黒丸で実現しよう！
 - クリックするだけで白丸の後は黒丸、黒丸の後は白丸を自動で続けるようにせよ
 - 他のコマが置かれている場所には置けないようにせよ
 - 考え方
 - Tic Tac Toe改をさらに改良するだけで良さそう！
 - 置こうとしている場所に、すでにコマが置かれていたら(値が1以上だったら)置けないようにする！



Tic Tac Toe 改二

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- そこに置かれているかどうかをチェックする！
 - 0なら置ける！ 1か2なら置けない！

```
void mousePressed(){  
    int tx = mouseX / 100;  
    int ty = mouseY / 100;  
    if( square[tx][ty] == 0 ){  
        square[tx][ty] = turn;  
        turn++;  
        if( turn > 2 ){  
            turn = 1;  
        }  
    } else {  
        // すでに置かれている場合は  
        // なにもしないけど置けません！  
        // と表示してもよい  
    }  
}
```


たったこれだけ！

勝敗判定をどうするかは
もう少し学んでから

[演習] Lights Out



- 横5マス、縦5マス の盤面を作り、そのマス目上をクリックするとクリックされたマス目の上下左右とそのマス目自身の色を反転させるLights Outを作る
 - まずは、クリックすると上下左右反転させるものを目指す





- 考え方
 - 5x5のマス目にそれぞれ黒色または黄色の正方形を描画する
 - 5x5の配列を作る
 - 配列の値が0の場合は黒色, 1の場合は黄色とする
 - マウスクリックされた座標がどのマス目に該当するかを計算する
 - そのマス目および、上下左右のマス目の値を反転させる
 - 0と1の値の反転は【 $\text{square} = 1 - \text{square}$ 】を使うと簡単！
だけど、もちろんif文で書いても良い

まず初期化！



- 2次元配列の定義

変数の型 [][] 変数名 = new 変数の型 [要素数][要素数];

```
int [][] square = new int [5][5];
```

```
void setup(){
```

```
    size( 500, 500 );
```

```
    int x=0;
```

```
    while( x<5 ){
```

```
        int y=0;
```

```
        while( y<5 ){
```

```
            square[x][y] = 0;
```

```
            y++;
```

```
        }
```

```
        x++;
```

```
}
```

```
}
```

```
while
```

5x5の2次元配列の定義

```
for
```

```
int [][] square = new int [5][5];
```

```
void setup(){
```

```
    size( 500, 500 );
```

```
    for( int x=0; x<5; x++ ){
```

```
        for( int y=0; y<5; y++ ){
```

```
            square[x][y] = 0;
```

```
        }
```

```
    }
```

```
}
```

多重ループで0に初期化

描画する！



- square[x][y] の値に応じて描画
 - 多重繰り返しで描画する

```
void draw(){  
    background( 255 );  
  
    int x=0;  
    while( x<5 ){  
        int y=0;  
        while( y<5 ){  
            if( square[x][y] == 0 ){  
                // 黒色四角形を描画  
            } else if( square[x][y] == 1 ){  
                // 黄色四角形を描画  
            }  
            y++;  
        }  
        x++;  
    }  
}
```

while

```
void draw(){  
    background( 255 );  
  
    for( int x=0; x<5; x++ ){  
        for( int y=0; y<5; y++ ){  
            if( square[x][y] == 0 ){  
                // 黒色四角形を描画  
            } else if( square[x][y] == 1 ){  
                // 黄色四角形を描画  
            }  
        }  
    }  
}
```

for



0か1かで描画内容を変更

マウスクリックでの変化

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- マウスクリックされた時に、そのクリックされた座標に応じて、どの配列の値を変更するかを考える！
 - 1マスが100ピクセルなので、 $mouseX \div 100$ や、 $mouseY \div 100$ の値(値を切り捨てたもの)が配列の添え字([]の中で指定する値)となる。また、その添え字の周辺が対象となる
 - tx, ty を求めて、そこの周辺を考える！



マウスクリックでの変化

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- マウスクリックされた時に、そのクリックされた座標に応じて、どの配列の値を変更するかを考える！
 - 1マスが100ピクセルなので、`mouseX / 100`や、`mouseY / 100`の値(値を切り捨てたもの)が配列の添え字([]の中で指定する値)となる。また、その添え字の周辺が対象となる

```
void mousePressed(){  
    // どこのマス目をクリックしたか  
    // マス目の大きさが100なので...  
    int tx = mouseX / 100;  
    int ty = mouseY / 100;  
  
    square[tx][ty] = 1 - square[tx][ty];  
    square[tx-1][ty] = 1 - square[tx-1][ty];  
    square[tx+1][ty] = 1 - square[tx+1][ty];  
    square[tx][ty-1] = 1 - square[tx][ty-1];  
    square[tx][ty+1] = 1 - square[tx][ty+1];  
}
```

求めた tx, ty の
周辺をそれぞれ0と1を反転させる
if 文で書いても良いがここでは
省略表記をしている

ArrayIndexOutOfBoundsException: 5


ArrayIndexOutOfBoundsException: -1

マウスクリックでの変化

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 上下左右を反転させようとするとエラーが出るのは配列の外を操作しようとしているから！
- つまり、 $tx-1$ や $ty-1$ 、 $tx+1$ や $ty+1$ が、配列の外になつていなかをチェックする必要あり！
 - if 文で判定してチェック！




```
void mousePressed(){  
    int tx = mouseX/100;  
    int ty = mouseY/100;  
  
    flag[tx][ty] = 1 - flag[tx][ty];  
    if( tx-1 >= 0 ){  
        flag[tx-1][ty] = 1 - flag[tx-1][ty];  
    }  
    if( tx+1 < 5 ){  
        flag[tx+1][ty] = 1 - flag[tx+1][ty];  
    }  
    if( ty-1 >= 0 ){  
        flag[tx][ty-1] = 1 - flag[tx][ty-1];  
    }  
    if( ty+1 < 5 ){  
        flag[tx][ty+1] = 1 - flag[tx][ty+1];  
    }  
}
```

このままでもよいけれど
判定が少々面倒…

[演習] Lights Out 改



- 【横5マス、縦5マス】の盤面を作り、そのマス目上をクリックするとクリックされたマス目の上下左右とそのマス目自身の色を反転させるLights Outを作る
- すべてが黒色になつたらCLEAR!と表示する





- 黄色のマスの数を数えて、合計が1以上だったらクリアではなく、合計が0だったらクリアと表示したら良い！
- やり方としては例えば下記の2つ
 - mousePressedの時に数を数えて、その数が0だったらclearというフラグ変数を0から1にし、0の場合はゲームを表示し、1の場合は「Clear!」という画面を表示する
 - drawの度に黄色の数を数えて、その数が0だったらそのままゲームを表示し、1以上だったら「Clear!」という画面を表示する

clear変数で管理

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- clearという変数を用意して、その変数の値を変化させる

```
int clear = 0;  
void mousePressed(){  
    // 反転とかもろもろの処理は省略  
  
    int total=0;  
    int x=0;  
    while( x<5 ){  
        int y=0;  
        while( y<5 ){  
            if( square[x][y] == 1 ){  
                total++;  
            }  
            y++;  
        }  
        x++;  
    }  
    if( total > 0 ){  
        clear = 1;  
    }  
}
```

while

```
int clear = 0;  
void mousePressed(){  
    // 反転とかもろもろの処理は省略  
  
    int total=0;  
    for( int x=0; x<5; x++ ){  
        for( int y=0; y<5; y++ ){  
            if( square[x][y] == 1 ){  
                total++;  
            }  
        }  
    }  
  
    if( total > 0 ){  
        clear = 1;  
    }  
}
```

for

clear変数で管理



- clear変数の値に応じて表示を切り替える

```
void draw(){  
    background( 255 );  
  
    if( clear == 0 ){  
        x=0;  
        while( x<5 ){  
            y=0;  
            while( y<5 ){  
                if( square[x][y] == 0 ){  
                    // 黒色四角形を描画  
                } else if( square[x][y] == 1 ){  
                    // 黄色四角形を描画  
                }  
                y++;  
            }  
            x++;  
        }  
    } else {  
        text( "Clear!", 100, 200 );  
    }  
}
```

while

```
void draw(){  
    background( 255 );
```

for

clearに応じて描画切り替え

drawに集約



```
void draw(){
    background( 255 );
    int total=0;
    int x=0;
    while( x<5 ){
        int y=0;
        while( y<5 ){
            if( square[x][y] == 1 ){
                total++;
            }
            y++;
        }
        x++;
    }

    if( total == 0 ){
        x=0;
        while( x<5 ){
            y=0;
            while( y<5 ){
                if( square[x][y] == 0 ){
                    // 黒色四角形を描画
                } else if( square[x][y] == 1 ){
                    // 黄色四角形を描画
                }
                y++;
            }
            x++;
        }
    } else {
        text( "Clear!", 100, 200 );
    }
}
```

while

数を数えて

数に応じて描画切り替え

```
void draw(){
    background( 255 );
    int total=0;
    for( int x=0; x<5; x++ ){
        for( int y=0; y<5; y++ ){
            if( square[x][y] == 1 ){
                total++;
            }
        }
    }

    if( total == 0 ){
        for( int x=0; x<5; x++ ){
            for( int y=0; y<5; y++ ){
                if( square[x][y] == 0 ){
                    // 黒色四角形を描画
                } else if( square[x][y] == 1 ){
                    // 黄色四角形を描画
                }
            }
        }
    } else {
        text( "Clear!", 100, 200 );
    }
}
```


for

Lights Out を別の解法で

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 5x5のマス目からはみ出した処理が面倒！
- 配列を拡張して、7x7のマス目を作って、その一部を表示する！（配列の溢れを防ぐため）
- 左端右端、上端下端の部分は表示せず、そこを除いた部分だけを表示する！



初期化！



- 表示領域の外も0で初期化してしまう（どうせ表示しないからなんでも良いけれど）

```
int [][] square = new int [7][7];
```

while

```
void setup(){
```

```
size( 500, 500 );
```

```
int x=0;
```

```
while( x<7 ){
```

```
int y=0;
```

```
while( y<7 ){
```

```
square[x][y] = 0;
```

```
y++;
```

```
}
```

```
x++;
```

```
}
```

```
}
```

5x5の2次元配列の定義

for

```
int [][] square = new int [7][7];
```

```
void setup(){
```

```
size( 500, 500 );
```

```
for( int x=0; x<7; x++ ){
```

```
for( int y=0; y<7; y++ ){
```

```
square[x][y] = 0;
```

```
}
```

```
}
```

多重ループで0に初期化

マウスクリックでの変化

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- マウスクリックされた時に、そのクリックされた座標に応じて、どの配列の値を変更するかを考える！
 - 変更するマス目は、それぞれ+1したもの！
 - tx と ty は1～5の値を取るので、-1や+1してもはみ出ない！

```
void mousePressed(){  
    // どこのマス目をクリックしたか  
    // マス目の大きさが100なので...  
    int tx = mouseX / 100 + 1;  
    int ty = mouseY / 100 + 1;  
    square[tx][ty] = 1 - square[tx][ty];  
    square[tx-1][ty] = 1 - square[tx-1][ty];  
    square[tx+1][ty] = 1 - square[tx+1][ty];  
    square[tx][ty-1] = 1 - square[tx][ty-1];  
    square[tx][ty+1] = 1 - square[tx][ty+1];  
}
```

描画するのはどこ？

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- square[x][y] の値に応じて描画
 - 多重繰り返しで描画する

```
void draw(){  
    background( 255 );  
  
    int x=1;  
    while( x<=5 ){  
        int y=1;  
        while( y<=5 ){  
            if( square[x][y] == 0 ){  
                // 黒色四角形を描画  
            } else if( square[x][y] == 1 ){  
                // 黄色四角形を描画  
            }  
            y++;  
        }  
        x++;  
    }  
}
```

while

```
void draw(){  
    background( 255 );  
  
    for( int x=1; x<=5; x++ ){  
        for( int y=1; y<=5; y++ ){  
            if( square[x][y] == 0 ){  
                // 黒色四角形を描画  
            } else if( square[x][y] == 1 ){  
                // 黄色四角形を描画  
            }  
        }  
    }  
}
```

for


0か1かで描画内容を変更

電光掲示板を作る

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 600x400のウインドウの上に、半径10の円を
敷き詰め、マウスカーソルが触れた場所が赤色
に灯る電光掲示板のようなものを作るには？



電光掲示板を作る

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



・考え方

- 600x400に半径10の円を敷き詰める場合、横に30個、縦に20個敷き詰めることが可能
- ライトが光っているか、光っていないかを管理する30x20の整数の配列を用意する
 - `int [][] light = new int [30][20];`
 - 光っていない場合は 0、光っている場合は 1 とする
- 円の中心の座標は、変数 i と j を利用した場合
($i*20+10, j*20+10$) となる(i, j は0から1ずつ増やす)
- $[i, j]$ 番目の円の中心からマウスまでの距離を求め
て中に入っていれば `light[i][j] = 1;` とする
- `light[i][j]` が 1 なら塗りつぶしを赤、0なら白とする

電光掲示板を作成する

```
int [][] light = new int [30][20];
void setup(){
    size( 600, 400 );
    // 最初にすべてを0にする
    int i=0;
    while( i<30 ){
        int j=0;
        while( j<20 ){
            light[i][j] = 0;
            j++;
        }
        i++;
    }
}
```

```
void draw(){
    background( 255, 255, 255 );
    int i=0;
    while( i<30 ){
        int j=0;
        while( j<20 ){
            // 円の中に入っていれば1にする
            if( dist(i*20+10, j*20+10, mouseX, mouseY)<10 ){
                light[i][j] = 1;
            }
            // 1なら赤色, 0なら白色にする
            if( light[i][j] == 1 ){
                fill( 255, 0, 0 );
            } else {
                fill( 255, 255, 255 );
            }
            // 中心(i*20+10, j*20+10)の円を描画
            ellipse( i*20+10, j*20+10, 20, 20 );
            j++;
        }
        i++;
    }
}
```

電光掲示板を作成する

```
int [][] light = new int [30][20];
void setup(){
    size( 600, 400 );
    // 最初にすべてを0にする
    for( int i=0; i<30; i++ ){
        for( int j=0; j<20; j++ ){
            light[i][j] = 0;
        }
    }
}
```

for でも書ける！

```
void draw(){
    background( 255, 255, 255 );
    for( int i=0; i<30; i++ ){
        for( int j=0; j<20; j++ ){
            // 円の中に入っていれば1にする
            if( dist(i*20+10, j*20+10, mouseX, mouseY)<10 ){
                light[i][j] = 1;
            }
            // 1なら赤色、0なら白色にする
            if( light[i][j] == 1 ){
                fill( 255, 0, 0 );
            } else {
                fill( 255, 255, 255 );
            }
            // 中心(i*20+10, j*20+10)の円を描画
            ellipse( i*20+10, j*20+10, 20, 20 );
        }
    }
}
```

数が明確な場合は
forがわかりやすいかも





- 電光掲示板のプログラムで、マウスでクリックされるとON/OFFを切り替えるようにしましょう
 - ONなら赤色、OFFなら白色
 - マウスクリック mousePressed の時に距離の判定を行なって、ON/OFFを切り替えるライトを探す
- 電光掲示板のプログラムで、マウスのクリック回数で色を変えるようにしてみましょう
 - (例) 白→赤→緑→青→白など

スクロールする電光掲示板

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



(Q) 600x400のウインドウの上に、半径10の円を
敷き詰めた電光掲示板を、右から左にスクロール
させていくにはどうするか？



スクロールする電光掲示板

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



・考え方

- 30x20の整数の配列 light を用意して、まずは全てを0に設定する
- すべての[i, j]について、マウスカーソルの下にある場合は、 $\text{light}[i][j] = 1;$ とする
 - $\text{dist}(i*20+10, j*20+10, \text{mouseX}, \text{mouseY})$
- $\text{light}[i][j]$ が0なら白色、1なら赤色に塗り色を設定
- [i, j]に相当する位置に円を描画する
- 一回 draw される毎に、光っているライトの位置を右から左へ移動する(1つ左に、1つ右のを代入)
 - $\text{light}[i-1][j] = \text{light}[i][j];$

スクロールする電光掲示板

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



lightの配列のある j について...

0	1	2	3
0	1	0	0

.....

27	28	29
0	1	1

.....



1番目を 2番目を 3番目を 4番目を
0番目に 1番目に 2番目に 3番目に



27番目を 28番目を 29番目を
26番目に 27番目に 28番目に

0

light[i-1][j] = light[i][j];

1つずつ左へずらしていく
1番右には 0を入れる

配列のエラーに注意

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- $\text{light}[i-1][j] = \text{light}[i][j]$ のため, i は1以上
 - i が0のとき, $\text{light}[-1][j] = \text{light}[0][j]$ となってしまい,
エラーになる! (-1番目なんて無い!)
- 右を左にコピーするので, 左のものから順に処理をしていく
 - i は順に1ずつ増やしていく
- 多重ループは j を外のループにして, i を内のループにしたほうがわかりやすい

```
for( int j=0; j<20; j++ ){  
    for( int i=1; i<30; i++ ){  
        light[i-1][j] = light[i][j];  
    }  
}
```



スクロールする

```
int [][] light = new int [30][20];
void setup(){
  size( 600, 400 );
  // 最初にすべてを0にする
  int i=0;
  while( i<30 ){
    int j=0;
    while( j<20 ){
      light[i][j] = 0;
      j++;
    }
    i++;
  }
}
```

```
void draw(){
  background( 255, 255, 255 );
  int i=0;
  int j=0;
  while( j<20 ){
    i=1;
    while( i<30 ){
      light[i-1][j] = light[i][j];
      i++;
    }
    light[29][j] = 0;
    j++;
  }
  i=0;
  while( i<30 ){
    j=0;
    while( j<20 ){
      if( dist(i*20+10, j*20+10, mouseX, mouseY)<10 ){
        light[i][j] = 1;
      }
      if( light[i][j] == 1 ){
        fill( 255, 0, 0 );
      } else {
        fill( 255, 255, 255 );
      }
      ellipse( i*20+10, j*20+10, 20, 20 );
      j++;
    }
    i++;
  }
}
```

スクロールする

```
int [][] light = new int [30][20];
void setup(){
  size( 600, 400 );
// 最初にすべてを0にする
for( int i=0; i<30; i++ ){
  for( int j=0; j<20; j++ ){
    light[i][j] = 0;
  }
}
}
```

```
void draw(){
  background( 255, 255, 255 );
  for( int j=0; j<20; j++ ){
    for( int i=1; i<30; i++ ){
      light[i-1][j] = light[i][j];
    }
    light[29][j] = 0;
  }
  for( int i=0; i<30; i++ ){
    for( int j=0; j<20; j++ ){
      if( dist(i*20+10, j*20+10, mouseX, mouseY)<10 ){
        light[i][j] = 1;
      }
      if( light[i][j] == 1 ){
        fill( 255, 0, 0 );
      } else {
        fill( 255, 255, 255 );
      }
      ellipse( i*20+10, j*20+10, 20, 20 );
    }
  }
}
```





- ・先述のプログラムのスクロール方向を左から右にしてみましょう
- ・先述のプログラムのスクロール方向を上から下にしてみましょう
- ・先述のプログラムで文字を表示してみましょう
 - どうやったら文字を表示できるでしょうか？

[演習] 左端から右端へ

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- ・ クリックによって赤／白の表示が切り替わる1次元の掲示板を、1秒毎に順に左側に移動し、左端のものを右端から登場させたい
- ・ 単純に繰り返しで移動すると右端が白になってしまう




[演習] 左端から右端へ



- 考え方

- 値を順に右から左へコピーしていく

- $\text{status}[0] = \text{status}[1];$
 - $\text{status}[1] = \text{status}[2];$
 - ⋮
 - $\text{status}[7] = \text{status}[8];$
 - $\text{status}[8] = \text{status}[9];$
 - ここで、 $\text{status}[9] = \text{status}[0];$ にするとすでに $\text{status}[0]$ には $\text{status}[1]$ の値が代入されてしまっているので駄目！




[演習] 左端から右端へ



- 考え方

- 左端の値を、一時的に他の変数に保存しておいて、そこに保存していた値を右端に入れれば良い！

- **temp = status[0];**
- **status[0] = status[1];**
- **status[1] = status[2];**
- :
- **status[7] = status[8];**
- **status[8] = status[9];**
- **status[9] = temp;**





```
int [] status = new int [10];  
  
void draw(){  
    int temp = status[0];  
    int x=0;  
    while( x < 9 ){  
        status[x] = status[x+1];  
        x++;  
    }  
    status[9] = temp;  
  
    x = 0;  
    while( x < 10 ){  
        if( status[x] == 1 ){  
            fill( 255, 0, 0 );  
        } else {  
            fill( 255 );  
        }  
        rect( x*50, 0, 50, 50 );  
        x++;  
    }  
}
```

while

```
int [] status = new int [10];  
  
void draw(){  
    int temp = status[0];  
    for( x=0; x<9; x++ ){  
        status[x] = status[x+1];  
    }  
    status[9] = temp;  
  
    for( int x=0; x<10; x++ ){  
        if( status[x] == 1 ){  
            fill( 255, 0, 0 );  
        } else {  
            fill( 255 );  
        }  
        rect( x*50, 0, 50, 50 );  
    }  
}
```

for




- ・右端のものを、左端から登場させるにはどうするか？
- ・2次元の掲示板のアニメーションを行い、左端から右端に登場させるようにするにはどうしたらよいだろうか？

[演習] 避けゲーム

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室




- 400x800のウインドウを作り,
Y座標が100, 200, 300,
400, 500, 600, 700のところ
に, それぞれ円を10個ずつ
ランダムに表示しましょう



円を一斉に動かす



(Q) 400x400の画面上の、Y座標100, Y座標200, Y座標300の場所に円を10個ずつ表示された円を左から右に動かそう(直径20)また、速度はランダムにせよ



円を一斉に動かす

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



・考え方

- 3x10個のX座標を格納する float 型の2次元配列を用意 (float [][] posX = new float [3][10];)
- 3x10個の速度を格納する float 型の2次元配列を用意 (float [][] speedX = new float [3][10];)
- 3x10個の配列に random で適当な値を代入
- 3x10個の配列の値に応じて円を描画
- 1回描画する度に、配列のそれぞれのX座標を speedX 分だけ右へ移動

円を一斉に動かす

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



- 3x10の2次元配列を作成し、それをそのまま変化させる！


```
float [][] posX = new float[3][10];
float [][] speedX = new float[3][10];
void setup(){
    size( 400, 400 );
    int i=0;
    while( i<3 ){
        int j=0;
        while( j<10 ){
            posX[i][j] = random(0,400);
            speedX[i][j] = random(5);
            j++;
        }
        i++;
    }
}
```

```
void draw(){
    background( 255 );
    fill( 0, 255, 0 );
    ellipse( mouseX, mouseY, 20, 20 );
    fill( 255, 0, 0 );
    int i=0;
    while( i<3 ){
        int j=0;
        while( j<10 ){
            posX[i][j] += speedX[i][j];
            ellipse( posX[i][j], (i+1)*100, 20, 20 );
            j++;
        }
        i++;
    }
}
```

円を一斉に動かす改



(Q) 先ほど作成した400x400の画面に3列に配置された10個の円を右方向に一斉に動く円は、右端に来ると左端から出てくるようにしよう



円を一斉に動かす改

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室



if(posX[i][j] > 400) なら posX[i][j] から 400引く！

```
float [][] posX = new float [3][10];
float [][] speed = new float [3][10];
void setup(){
    size( 400, 400 );
    int i=0;
    while( i<3 ){
        int j=0;
        while( j<10 ){
            posX[i][j] = random(0,400);
            speed[i][j] = random(0,5);
            j++;
        }
        i++;
    }
}
```

```
void draw(){
    background( 255 );
    fill( 0, 255, 0 );
    ellipse( mouseX, mouseY, 20, 20 );
    fill( 255, 0, 0 );
    int i=0;
    while( i<3 ){
        int j=0;
        while( j<10 ){
            posX[i][j] = posX[i][j] + speed[i][j];
            if( posX[i][j] > 400 ){
                posX[i][j] = posX[i][j] - 400;
            }
            ellipse( posX[i][j], (i+1)*100, 20, 20 );
            j++;
        }
        i++;
    }
}
```



- 一斉に動く円との衝突判定をしてみましょう
- 一斉に動く円の数を、10個から20個に増やしてみましょう
- 一斉に動く円の数を、10個から100個に増やしてみましょう
- 一斉に動く円の方向を左方向にしてみましょう
 - speed をマイナスにしたらOK！
- 一斉に動く円を、右方向にも左方向にも動くようにしてみましょう
 - speed を-5から5までにしたらOK！